	Das Codereview

Alf Borrmann

Mit dem „Codereview“ wird eine neue Reihe gestartet, die für den Programmierer in der Praxis, also Sie, gedacht ist. Die hier gezeigten Programme sind mir bei meiner Praxis in verschiedenen Projekten aufgefallen. Sie wurden bzw. werden tatsächlich in Applikationen eingesetzt, wo sie Probleme teilweise bereits verursacht haben oder befürchten lassen. Ich habe die Programme auf vorhandene oder potenzielle Probleme aber auch auf gute Lösungen und pfiffige Ideen untersucht. Meine Überlegungen zu besseren Codestrukturen oder anderen Programmierstrategien stelle ich Ihnen hier vor. Ich nehme dabei nicht in Anspruch, die jeweils ideale Lösung zu zeigen, sondern will Ihnen Hinweise auf verschiedene VFP-Befehle, vielleicht unbekannte Parameter und ihre Auswirkungen sowie Programmierstile geben.


Gerne können Sie mir Ihre Meinung zu den vorgestellten Konstrukten oder Verbesserungen hierzu mitteilen, ich werde diese dann in der jeweils nächsten Folge des „Codereviews“ aufgreifen.

Sollten Sie ein Programm haben, das Ihnen suspekt erscheint, ein Problem hat oder eins, das Ihrer Meinung nach eine besonders clevere Problemlösung darstellt, schreiben Sie mir (Alf.Borrmann@denk-modell.de), um dieses Programm (auf Wunsch anonym) besprechen zu lassen.

So, nun aber ran an das erste „corpus delicti“. Dies sind in dieser Folge zwei Programme, die ich, um sie besser abdrucken zu können, leicht gekürzt habe, ich habe aber den Originalcode nicht in für das Review relevanter Weise verändert, die Zeilennummern habe ich nur für die Referenzierung hinzugefügt.

Das erste ist ein Programm namens „ReUSE“. Dieses Programm soll folgende Problematik lösen: wenn man mit einem SQL-Select einen cursor erstellt hat, so kann dieser gut als Quelle für eine Auswahlliste in Form einer Listbox auf dem Bildschirm dienen. Ein per SQL-Select erstellter cursor hat immer eine physikalische Reihenfolge der Datensätze, die sich (wenn nicht anders in der „order“-Klausel angegeben) immer an den aufsteigend sortierten Werten des ersten Feldes orientiert. Da die Listbox mehrere Spalten haben kann, die jeweils einzelnen Feldern des Cursors entsprechen, besteht ggf. der Wunsch der Benutzer, die Listbox jeweils nach verschiedenen Spalten sortieren zu lassen. Nun: nichts leichter als das, mag man meinen, man kann ja einfach einen Index für die in Frage kommenden Felder aufbauen und diesen durch den Benutzer wählen lassen. Diese Technik ist sehr mächtig, denn VFP indiziert Tabellen bis zu ca. 1000 Datensätzen dermaßen schnell, daß der Benutzer davon normalerweise nichts mitbekommt.

Der Code hierfür sähe etwa so aus:

select <Felder> ... into cursor temp

index on <Feld1> tag <Tag1>

index on <Feld2> tag <Tag2>

...

Leider erhält man bei der zweiten „index on“-Anweisung die Fehlermeldung „Datei <temp> ist schreibgeschützt“. Tatsächlich ist es so, daß die Abfrageergebnisse von VFP in nicht beschreibbaren temporären Dateien abgelegt werden. Spätestens mit der zweiten Anweisung versucht VFP aber eine .CDX-Datei zu der Tabelle anzulegen. Die Information, dass zu dieser Tabelle nun eine .CDX-Datei besteht, schreibt VFP in den Header der Tabelle, was bei schreibgeschützten Dateien natürlich zu einem Fehler führt. Eine Lösung hierfür ist, den Cursor in einem zweiten Arbeitsbereich ein weiteres Mal zu öffnen. Er ist dann nicht mehr schreibgeschützt, es lassen sich also beliebige Indizes anlegen. Dieses Problem also löst das Programm ReUSE:

 100
* Procedure...:
ReUSE.prg

 200
* Parameters..:
tcAlias: Alias des Cursors als String

 400
* Author:.....: <anonymisiert>

 500
* History.....:
29.07.97

 700
parameters tcAlias

1000
private lcOldSelect, lcTempAlias, i

1100
lcOldSelect = select(0)

1200
lcTempAlias = sys(2015)

1400
if right(tcAlias,3)=="TMP“

1500
* echter Cursor, kein interner Filter auf die Original-DBF

1700

use (dbf(tcAlias)) alias &lcTempAlias in 0 again

1800
else
&& Filter oder Projektion der Original-DBF

2000

*-- Arbeitsbereich des uebergebenen Cursor waehlen

2100

select (tcAlias)

2300

*-- Struktur des uebergebenen Cursors in Array lesen

2400

private laTempArray

2500

=afields(laTempArray)

2700

*-- Struktur des Arrays pruefen.

2800

*-- Numerische Felder duerfen maximal 20 Stellen haben.

2900

*-- Nachkommastellen werden auf 10 begrenzt.;





 Mehr macht in der Regel keinen Sinn.

3000

*!* FOR i=1 TO ALEN( laTempArray, 1 )

3100

*!* 
IF laTempArray[i,2] = "N" AND laTempArray[i,3] > 20

3200

*!* 

laTempArray[i,3] = 20

3300

*!* 

laTempArray[i,4] = MIN( laTempArray[i,4], 10 )

3400

*!* 
ENDIF

3500

*!* ENDFOR

3700

*-- neuen Cursor mit derselben Struktur in einem;




 freien Arbeitsbereich erstellen

3900

select 0

4000

create cursor (lcTempAlias) from array laTempArray

4200

*-- Daten aus dem temporaeren Alias an den neuen Cursor uebergeben

4300

append from dbf(tcAlias)

4400
endif

4500


4600
*-- Arbeitsbereich des uebergebenen Cursor waehlen

4700
select (tcAlias)

4900
*-- den uebergebenen Cursor schliessen

5000
use

5100


5200
*-- jetzt wird der uebergebene Cursor in seinem;



 urspruenglichen Arbeitsbereich

5300
*-- und unter seinem urspruenglichen Namen wieder geoeffnet

5400
use (dbf(lcTempAlias)) alias &tcAlias shared again

5500


5600
*-- temporaeren Alias wieder schliessen

5700
use in (lcTempAlias)

5800


5900
*-- urspruenglichen gewaehlten Arbeitsbereich wieder waehlen

6000
select (lcOldSelect)

6100


6200 return

Dieses Programm stammt offenbar aus einer Version vor VFP 3.0, da es von verschiedenen Features keinen Gebrauch macht. Zunächst wäre hier die Deklararion der Variablen und Parameter zu nennen. Unter dem Gesichtspunkt der Codehygiene sollten Variablen und Parameter immer „local“ deklariert werden. Damit sind sie nur in dieser Prozedur bzw. Methode und nicht mehr in ggf. aufgerufenen Prozeduren sichtbar. Daneben ist diese Deklaration etwas schneller als die Deklaration als „private“ (s. a. meinen Vortrag auf der DevCon '99). Dies geht für Parameter mit dem Statement „lparameter“ statt „parameter“, Variablen werden mit „local“ statt „private“ deklariert. Dabei sollte man sich an die in der VFP-Hilfe vorgeschlagenen Namenskonventionen halten, also erster Buchstabe für Sichtbarkeit („l“ für local, „p“ für private, „g“ (global) für public und „t“ für Parameter) und zweiter Buchstabe für Datentyp („c“ für Character, „n“ für numerisch, „a“ für Array etc.). Die Deklaration sollte also heißen:

local lcOldSelect, lcTempAlias; lnI

Noch etwas zur Parameterprüfung: sollte der Nutzer dieser Funktion einen Parameter übergeben, der keinen gültigen Alias enthält oder das Übergeben eines Parameters gänzlich unterlassen, erzeugt die Funktion einen Laufzeitfehler. Um dies beim Test zu erkennen, können Sie sehr gut die Anweisung „assert“ (neu in VFP 5.0) einsetzen. Sie wertet einen logischen Ausdruck aus und öffnet bei „logisch falsch“ dieses Ausdrucks einen Dialog, der es dem Entwickler erlaubt, den Debugger aufzurufen. Besonderer Vorteil dieses Befehls: er kostet in der Laufzeitumgebung keine Zeit, weil er dort ignoriert wird, mithin die Laufzeit des ausgelieferten Programms nicht beeinträchtigt. Die Anwendung des „assert“ sähe hier wie folgt aus:

assert pcount( ) = 1;

 and vartype( tcAlias) = „C“;

 and used( tcAlias);

message “Es muß der Alias einer geöffneten Tabelle übergeben werden.”

Die nächste Zeile liefert ein weiteres potenzielles Problem. Die Funktion „select( 0)“ liefert nicht den Namen, sondern die Nummer des aktuellen Arbeitsbereichs. Um also im weiteren Programmverlauf den Typ der Variable sehen zu können, sollte sie „lnOldSelect“ heißen.

In der nächsten Zeile wird eine meiner Lieblingsfunktionen genutzt: „sys( 2015)“. Diese VFP-Funktion liefert einen immer wieder neu berechneten String aus 10 Zeichen. Der läßt sich sehr gut einsetzen, um Tabellen mit einem temporären Alias zu öffnen. Der Name des Alias muß dann nicht mehr per Hand im Code fest angelegt werden, wodurch ein geschachteltes Aufrufen von Prozeduren und Methoden möglich ist (Stichworte Polymorphie und Kapselung), ohne daß auf die falsche Tabelle zugegriffen wird.

In der Zeile 1400 wird geprüft, ob die Datei eine temporäre Datei ist. Die hier verwendete Funktion „right( ..., 3) wird angewendet, um die Dateiendung abzufragen. Das ist in diesem Falle statthaft, da VFP Cursor immer mit der Endung „.TMP“ anlegt. Eleganter und sicherer ist hierfür aber die nun aus den Foxtools in VFP integrierte Funktion „justext( )“, die immer die Dateiendung liefert (man stelle sich vor, man untersucht eine Datei, die mehr oder weniger als 3 Zeichen als Endung hat).

Danach wird die Datei einfach mit Hilfe des Befehls „use ... again“ ein zweites Mal geöffnet. Dies ist seit der FoxPro-Version 2.0 möglich. Damit kann man – ohne ein weiteres Dateihandle zu verwenden, also ohne zusätzlichen Laufzeitoverhead des Betriebssystems - eine Tabelle ein zweites Mal öffnen. Der „use“-Befehl hat über die Zeit ein paar Klauseln hinzu bekommen, die vielleicht der Erläuterung bedürfen. Zunächst die Klausel „in <Arbeitsbereich>“. Hier können Sie – ohne vorheriges „select <Arbeitsbereich>“ - einen Bereich auswählen, in dem die Tabelle zu öffnen ist. Das im use-Befehl eingebaute Wechseln in den angegebenen (und Zurückwechseln in den vorher gewählten) Bereich spart dabei Code und Laufzeit. Geben Sie hier „in 0“ an, verwendet VFP den nächsten freien Bereich. Zusätzlich empfiehlt es sich immer, anzugeben, ob die Tabelle exklusiv oder nicht exklusiv geöffnet werden soll. In dem hier vorliegenden Fall kann das zwar unterbleiben, da ein Cursor von VFP immer exklusiv geöffnet wird und von daher auch ein weiteres Mal exklusiv geöffnet werden kann. Das „exclusive“ von Visual FoxPro bezieht sich nämlich auf den Zugriff der aktuellen Instanz von VFP im Speicher auf die physikalische Tabelle auf der Platte. Für die Codehygiene kann die Klausel „shared“ alternativ zu „exclusive“ verwendet werden. Sie öffnet die Tabelle wie gewünscht, ohne daß die Einstellung von „set exclusive ON|OFF“ eine Rolle spielt. Damit wird das Programm unabhängig von der Einstellung.

Interessant in der Zeile 1700 ist auch die Verwendung des Makros: Makroexpansionen sollten immer mit einem Punkt abgeschlossen werden, also „... alias &lcTempAlias. ...“. Insgesamt ist die Verwendung eines Makros in der „alias“-Klausel aber unnötig, eine einfache Klammerung reicht. Wichtig zu wissen ist dies in dem Fall, daß der zu verwendende Alias nicht in einer Variable, sondern im Property eines Objektes vorliegt, denn diese lassen sich nicht per Makroexpansion erweitern. Die komplette Zeile zum Öffnen der Tabelle könnte also so aussehen:

use ( dbf( tcAlias)) alias ( lcTempAlias) shared in 0 again

Der “else”-Teil der „if“-Anweisung ab Zeile 200 ist besonders interessant: hier wird der Fall behandelt, daß in „tcAlias“ eine normale Tabelle (oder zumindest keine „.TMP“-Datei) geöffnet ist. Das kann dann passieren, wenn ein SQL-„select“-Befehl lediglich einen Ausschnitt aus einer tatsächlich vorhandenen Tabelle betrifft, der die gleiche Struktur wie die Tabelle hat. In dem Fall öffnet VFP nämlich die Tabelle intern mit der gleichen Strategie wie in dem „use ... again“-Befehl beschrieben und setzt auf den Alias einen Filter, so daß nur die von der where-Klausel des SQL-selects betroffenen Datensätze sichtbar sind. Sollte nun das Programm „ReUSE“ dafür verwendet werden, einen temporären Index per „set order to ..“ zu erstellen, so würde dieser in dem Falle jedoch nicht temporär angelegt, sondern in der .CDX-Datei zu der Originaltabelle gespeichert und dort erhalten bleiben. Um dies zu vermeiden, wird im Falle, daß es keine „.TMP“-Datei ist, angenommen, daß es eine gefilterte Tabelle ist und der Inhalt in eine neue Datei kopiert. Dies wird in den Zeilen 2000 – 4300 recht geschickt über die Funktion „afields( )“ gemacht. Diese kopiert die Struktur (Feldnamen, Feldtyp, Feldbreite, Dezimalstellen, sowie 12 weitere Informationen) der Tabelle in ein Array. Aus der Information in diesem Array kann man dann eine neue Tabelle mit gleicher Struktur erstellen. Wie die meisten anderen Befehle oder Funktionen, die sich auf Tabellen beziehen, akzeptiert die Funktion „afields()“ inzwischen einen weiteren  Parameter, in dem der Alias der Tabelle angegeben werden kann. Das vorherige „select“ (Zeile 2000) ist also überflüssig.

In den Zeilen 2700 – 3700 war nun eine extrem gefährliche (um nicht zu sagen schmuddelige) Operation enthalten: die Struktur der Tabelle wurde verändert. Gefährlich ist dies deshalb, weil das Programm ja nicht weiß, welche Tabelle übergeben wurde und ob das aufrufende Programm nicht darauf angewiesen ist, die Tabelle exakt wie angegeben ein zweites Mal zu öffnen. Das hat der Autor dann auch gemerkt (möglicherweise hat das Programm schon einen Fehler verursacht) und die Zeilen auskommentiert. Dabei vergessen hat er aber, die Deklaration der Laufvariable „i“ (Zeile 100) zu löschen. Das ist zwar kein großer Beinbruch, aber es kostet eben Speicherplatz und Laufzeit. Aus Gründen der Codehygiene sollte also der gesamte Block 2700 – 3700 sowie die Deklaration der Variable komplett gelöscht werden.

In Zeile 4000 f. wird aus den Strukturinfos nun ein eigener Cursor erzeugt. Es ist hier nicht möglich, die neue Tabelle per „copy structure“ zu erstellen, da ein Cursor und keine neue Tabelle erstellt werden soll. Der „copy structure“-Befehl enthält aber keine Klausel, um einen Cursor zu erstellen. Warum dann überhaupt einen Cursor erstellen, immerhin kostet das den zusätzlichen Code für das Array? Nun: Cursor haben den Vorteil, daß Visual FoxPro sie, wenn sie geschlossen werden, mitsamt aller zugehöriger Dateien (.FPT und .CDX) wieder von der Platte löscht. Das muß also nicht per Hand erfolgen.

Nach dem Erzeugen einer neuen Tabelle werden die Datensätze aus dem übergebenen Alias an den neuen Cursor angehängt und sind damit also unabhängig von der Originaltabelle verfügbar.

Der Rest hinter dem „endif“ ist Routine: die Tabelle in dem übergebenen Alias wird geschlossen und der neu im Alias „lcTempAlias“ enthaltene Cursor wieder mit „tcAlias“ geöffnet. Danach der nur innerhalb von ReUSE benötigte Arbeitsbereich „tcTempAlias“ geschlossen. Warum diese Reihenfolge? Wie schon gesagt, löscht VFP Cursor von der Platte, sobald sie nicht mehr geöffnet sind. Würde also nun der Cursor in „lcTempAlias“ geschlossen, würde die Datei sofort gelöscht und könnte nicht mehr in „tcAlias“ geöffnet werden.

Ganz zum Schluß wird wieder der vorher aktive Arbeitsbereich gewählt, um die Umgebung wieder herzustellen.

Der Kommentar zu dem gesamten Programm: es könnte schöner programmiert sein, erfüllt aber alles in allem seinen Zweck, nämlich einen Cursor beschreibbar zu öffnen. Trotzdem halte ich das Programm für komplett verzichtbar, und zwar aus folgenden Gründen:

Insgesamt kostet das Programm mit allen seinen Anweisungen viel zuviel Laufzeit. Besonders der Teil, in dem die Daten komplett umkopiert werden (das können ja durchaus ein paar tausend Datensätze sein) verschlingt Zeit und ist nur dann sinnvoll, wenn auf eine in der Datenbank persistent existierende Tabelle ein zusätzlicher, nur temporär verwendeter Index gebraucht wird. Das Wissen, welche Art von Tabelle in „tcAlias“ enthalten ist, hat aber das aufrufende Programm, die Abfrage auf die Endung „.TMP“ ist also potentiell überflüssig. Das aufrufende Programm kann also auch die Operationen auf diese Tabelle erledigen, ohne dies an ReUSE zu delegieren.

Insbesondere unter dem Aspekt der objektorientierten Programmierung kann ein Objekt, das eine spezielle Sortierfolge in einer Tabelle benötigt, selbst entscheiden, welche Strategie hierfür zu verwenden ist. Grundgedanke ist hier, daß ein Objekt eine Tabelle (oder einen Cursor) selbst verwaltet. Hierbei weiß der Programmierer des Objektes (bzw. der Klasse) in der Regel, ob eine Tabelle als tatsächliche Tabelle oder eben als Cursor vorliegt.

So kann mit den folgenden drei Zeilen der gleiche Effekt für Cursor programmiert werden:

use ( dbf( <ggf. alter Alias>));

 alias <neuer Alias>;

 shared again in 0

use <ggf. alter Alias>

select <neuer Alias>

Dies spart den Aufruf einer separaten Funktion wie ReUSE (allein der Aufruf der Funktion kostet Zeit) sowie die Zeit zum Erstellen von Variablen, die „if“-Statements usw.

In dem Fall, daß ein Objekt die Tabelle verwaltet, ist auch das Erstellen eines temporären Indexes auf eine vorhandene persistente Tabelle kein Problem mehr: es wird einfach eine neue Indexdatei angelegt. Wird das Objekt gelöscht, kann dieses im „destroy“-Event des Objektes ebenfalls wieder gelöscht werden. Das Problem, daß hier weitere Indexe in der Datenbank angelegt werden, wäre damit gelöst:

function init

*
Tabelle wird geöffnet

index on <Feldname oder Ausdruck>;

 to <temporärer Indexdateiname>;

 compact additive

function destroy

*
Tabelle wird geschlossen

delete file <temporärer Indexdateiname>

Daneben kann das Objekt natürlich auch mehrere temporäre Indexdateien verwalten und somit bei Bedarf auf verschiedene Sortierfolgen für die Tabelle zugreifen.

Sollten diese Indexe allerdings oft benötigt werden, so empfiehlt sich sicherlich die Aufnahme in die Datenbank bzw. die .CDX-Datei, denn auch die zuletzt gezeigte Technik kostet natürlich Laufzeit für das Erstellen der Indexbäume und das Löschen der Dateien.

Es ist hier aber sehr gut zu sehen, daß innerhalb einer OO-Programmierung weniger Abfragen über den Zustand erfolgen müssen und damit die (in VFP recht langsamen) „if“-Statements entfallen können. Durch ordentlich strukturierte Vererbung kann diese Funktionalität an alle Objekt vererbt werden, die diese Funktionalität benötigen, der Code muß dadurch nicht mehrfach erstellt werden. Das bedeutet also, daß hierbei wesentlich weniger Code vonnöten ist, das Programm besser wartbar und auch schneller ist.

Haben Sie Fragen oder Anmerkungen zu den hier besprochenen Techniken? Haben Sie eine eigene Problemstellung? Schreiben Sie an den Autor.

Alf Borrmann ist Gesellschafter der denk-modell Software in Mainz und erreichbar per Alf.Borrmann@denk-modell.de

